Processingでシャルルの「夜の群れ」を作ってみた

この記事は[U-TOKYO AP Advent Calendar 2018](https://qiita.com/advent-calendar/2018/ut_ap)の24日目の記事です．他の皆さんが理論に立脚した良記事を書く中で恐縮ですが，僕は理論も何もない一芸を披露します．

<!-- TODO 動画貼り付け -->

# はじめに

皆さん今日が何の日かご存知でしょうか？そう，今日はupd8所属のバーチャルシンガー[YuNi](https://www.youtube.com/channel/UCHTnX0CSX_KObo5I9WuZ64g)の，VARKでのVRライブ[「YuNi 1st VR LIVE! 〜VeRy Merry X’mas〜」](https://vark.co.jp/2018/11/25/%E3%80%90%E9%87%8D%E8%A6%81%E3%80%91yuni-vr%E3%83%A9%E3%82%A4%E3%83%96%E3%83%81%E3%82%B1%E3%83%83%E3%83%88%E7%99%BA%E5%A3%B2%E6%97%A5%E6%B1%BA%E5%AE%9A%EF%BC%81/)の日ですね！用意されたチケットはわずか7分で売り切れたそうです．

<img src="https://qiita-image-store.s3.amazonaws.com/0/212951/329204fb-7547-d61e-b1be-47808a265164.png" width = 50%>


さて，YuNiがYouTubeに投稿している動画のうち，2018年12月23日正午時点で最も再生回数が多い動画は，[「シャルル」のカバー動画](https://youtu.be/LjkDJHk2KrE)です．その再生回数は443万回．聴いたことがない人は今すぐリンクを踏んで聴きましょう．「シャルル」自体聴いたことがない人も多いと思いますが，以下の話はシャルルのオリジナルPVを前提として書くので，この機会に[YouTubeのページ](https://youtu.be/TA5OFS_xX0c)で聴いてください．この曲は[バルーン](https://twitter.com/balloon0120)さん作曲のボカロ曲で，[アボガド6](https://twitter.com/avogado6)さんが映像を作成しています．2016年に公開され，2018年12月23日正午時点では再生回数1,079万回，[バルーンさんのセルフカバー版](https://youtu.be/VUIEJu4ZSUo)の再生回数はなんと2,047万回にも及んでいます．~~僕も今夏デスマーチを繰り広げているときに後輩がやたら流すのを聴いてハマりました．~~

この記事では特にPVの映像表現に着目します．具体的には，1回目のサビの「夜の群れ」部分です．[オリジナル動画の1:09付近](https://youtu.be/TA5OFS_xX0c?t=69)ですね．[YuNiのカバー動画の1:09付近](https://youtu.be/LjkDJHk2KrE?t=69)でもバッチリ再現されています．「夜」という文字が背景を暗く埋め尽くす表現は，初めてPVを見たときはかなり驚きました．今回はこの表現を**Processing**で実装してみます．

# 文字の生成
この表現は端的に，「ある文字が上から下にかけてランダムに生成されていく表現」と言えそうです．手始めにこの方針で実装してみます．

文字の位置や大きさ等といった情報を保持するクラス`Character`を定義して，これを元に文字を生成します．この際，現在のフレーム数`frameCount`を参照して，文字を生成する高さを制御します．

```java:Character.pde
class Character{
  String text;
  PVector position;
  color col;
  float size;
  
  Character(String _text, PVector _position, color _col, float _size){
    text = _text;
    position = _position;
    col = _col;
    size = _size;
  }
  
  void draw(){
    textAlign(CENTER);
    textSize(size);
    fill(col);
    text(text, position.x, position.y);
  }
}
```

```java:sharuru.pde
final color BACK = color(248, 242, 225);
boolean isPlaying = false;

void setup() {
  size(640, 360, P2D);
  background(BACK);
  fill(0);
  noStroke();
  frameRate(240);
  PFont font = createFont("HiraMinPro-W3", 160, true);
  textFont(font);
}

void draw() {
  if(isPlaying){
    Character character = new Character("夜", new PVector(random(0, width), random(frameCount, frameCount + 160)), color(43, 65, 75), random(80, 160));
    character.draw();
  }
}

void mouseReleased(){
  if(isPlaying){
    background(BACK);
  }else{
    frameCount = 0;
  }
  isPlaying = !isPlaying;
}
```

着目すべきポイントは，Processingのデフォルトのフォントだと「夜」という漢字が表示されないため，文字のフォントを日本語対応のものに変えている点です．

```java
PFont font = createFont("HiraMinPro-W3", 160, true);
textFont(font);
```

フォントは`PFont`型で定義されます．`createFont(String name, float size, boolean smooth)`関数は，`name`という名前のフォントを，`size`の大きさで使えるように，`PFont`型のインスタンスを生成します．また`smooth`を`true`にすると，アンチエイリアシングが働きます．ここでは`name`をヒラギノ明朝[^1]に，`size`を生成される可能性のある最大サイズに指定しています．そして`textFont()`関数でテキストのフォントを指定します．

このようにすると，マウスクリックを離したときにアニメーションが始まります．実際に動かすと…

![first.gif](https://qiita-image-store.s3.amazonaws.com/0/212951/4111117d-cbc1-f374-53b6-fdb9b4486db8.gif)

と微妙な出来栄えのものになりました．

# 改善策
今作ったものと原作を見比べると，次の3つのことに気が付きました．

* 今作ったものではところどころ隙間が残ってしまっているが，原作では最終的に隙間が綺麗に埋まっている．
* 原作では同じタイミングで複数個の文字を生成している．
* 原作では文字の生成範囲のうち下の方の密度はあまり高くなく，「夜」という文字が読みやすい．

これらに関して改善策を考えると，**画面全体をいくつかの範囲に等分し，1フレームで1つの範囲あたりに描画する文字数を固定する．そして毎フレーム描画対象になる範囲の数を増やしていく．それでも埋まらない隙間は，少しの遅延を入れて上からベタ塗りして埋める**という~~なんとも夢のない~~結論に至りました．概念図にすると次の通りです．

![image.png](https://qiita-image-store.s3.amazonaws.com/0/212951/000712f1-e286-3208-ff39-cac040cf6e1c.png)


この考えを元にしてコードを書き直します．

```java:sharuru.pde
final color BACK = color(248, 242, 225);
final int NUM = 10; // 1つの領域に一度に描画する文字数
final int COUNT = 8; // 画面分割数
final int DELAY = 4; // ベタ塗りのために遅らせるフレーム数

boolean isPlaying = false;

void setup() {
  size(640, 360, P2D);
  background(BACK);
  fill(0);
  noStroke();
  frameRate(10);
  PFont font = createFont("HiraMinPro-W3", 160, true);
  textFont(font);
}

void draw() {
  if(isPlaying){
    for(int i = 0; i < frameCount; i++){
      for(int j = 0; j < NUM; j++){
        Character character = new Character("夜", new PVector(random(0, width), random(i * (height / COUNT), i * (height / COUNT) + 160)), color(43, 65, 75), random(80, 160));
        character.draw();
      }
    }
    rect(0, (frameCount - DELAY) * (height / COUNT), width, height / COUNT);
  }
}

void mouseReleased(){
  if(isPlaying){
    background(BACK);
  }else{
    frameCount = 0;
  }
  isPlaying = !isPlaying;
}
```

# 完成物

<blockquote class="twitter-tweet" data-lang="ja"><p lang="ja" dir="ltr">Processingでシャルルの「夜の群れ」を作ってみた <a href="https://t.co/TY40NCZH6O">pic.twitter.com/TY40NCZH6O</a></p>&mdash; らでぃっしゅ (@radi_bow) <a href="https://twitter.com/radi_bow/status/1076693852306563072?ref_src=twsrc%5Etfw">2018年12月23日</a></blockquote>
<script async src="https://platform.twitter.com/widgets.js" charset="utf-8"></script>

![second.gif](https://qiita-image-store.s3.amazonaws.com/0/212951/70e91b2d-0a05-1302-36cd-0dbd8c17726c.gif)

GIFの方はフレームレートが落ちてしまっているのですが，動画の通り実際は原作とほぼ同じ速さで背景が塗りつぶされます．原作と完全に一致とは言えないと思いますが，それなりの完成度にはなっているのではないでしょうか．

---

さて，僕の卒業した計数工学科数理情報工学コースの授業でProcessingを習うことはないのですが，普段オブジェクト指向プログラミング，特にUnityでC#を使っている人であれば紹介したコードの雰囲気がすぐに掴めたと思います．こういう可視化ツールとしてのプログラミングを面白いと思った人は，是非**Creative Coding**の沼に足を踏み入れてみてください．

*それではVeRy Merry X’mas!!*

[^1]: MacではなくWindowsを使っていてヒラギノ明朝がPCに入っていない場合は，MSP明朝等を指定するとそれっぽく見えて良いと思います．